Οσοι δεν μπορούν να δουν τους Ελληνικούς χαρακτήρες παραθέτω τον κώδικα του προγράμματος typedf2.pas στη συνέχεια.

Το παρακάτω πρόγραμμα είναι αντίστοιχο μ' αυτό που σας έδωσα σε φωτοτυπία με τις εξής αλλαγές:

1. Σε κάθε διαδικασία ανοίγω και κλείνω το αρχείο. Καλύτερη τεχνική και πιο ασφαλής σε περιπτώσεις που δεν υπάρχει το αρχείο.

2. Η διαδικασία getData είναι λίγο διαφορετική. Η υλοποίηση της getData σ' αυτή την έκδοση του προγράμματος είναι ορθότερη. Κάνει χρήση της filesize(f) και της seek(f, filesize(f));. H filesize(f) είναι συνάρτηση που επιστρέφει το μέγεθος του αρχείου σε αριθμό εγγραφών. Ετσι αν ένα αρχείο έχει 10 εγγραφές, δηλαδή εγγραφές με αύξοντα αριθμό εγγραφής από 0 έως 9, η seek(f, filesize(f)); θα επιστρέψει 10 που σημαίνει ότι οδηγεί το δείκτη ανάγνωσης-εγγραφής του αρχείου στην 10η εγγραφή που είναι κενή εγγραφή, δηλαδή δεν έχουν καταχωρηθεί δεδομένα και άρα η προσθήκη των εγγραφών θα γίνει από εκεί και πέρα.

program typedf2;

uses crt;

type

    Friend=record

                 name:string[20];

                 address:string[15];

                 phone:longint;

    end;

var

   f :file of Friend;

   choice :char;

procedure menu(var reply:char);

begin

     clrscr;

     writeln('1. ΕΙΣΑΓΩΓΗ ΣΤΟΙΧΕΙΩΝ ΦΙΛΩΝ');

     writeln('2. ΕΜΦΑΝΙΣΩΝ ΟΛΟΥ ΤΟΥ ΑΡΧΕΙΟΥ');

     writeln('3. ΑΝΑΖΗΤΗΣΗ & ΕΜΦΑΝΙΣΗ ΜΕ ΒΑΣΗ ΤΟΝ ΑΡΙΘΜΟ ΕΓΓΡΑΦΗΣ');

     writeln('4. ΑΝΑΖΗΤΗΣΗ ΜΙΑΣ ΕΓΓΡΑΦΗΣ ΜΕ ΒΑΣΗ ΤΟΝ Α.ΕΓΓΡΑΦΗΣ & ΑΛΛΑΓΗ ΤΟΥ ΠΕΡΙΕΧΟΜΕΝΟΥ ΤΗΣ');

     writeln('5. ΑΝΑΖΗΤΗΣΗ ΜΙΑΣ ΕΓΓΡΑΦΗΣ ΜΕ ΒΑΣΗ ΤΟ ΟΝΟΜΑ & ΑΛΛΑΓΗ ΤΟΥ ΠΕΡΙΕΧΟΜΕΝΟΥ ΤΗΣ');

     writeln('0. ΤΕΛΟΣ');

     write('ΔΩΣΕ ΤΗΝ ΕΠΙΛΟΓΗ ΣΟΥ 0..5 ');

     repeat

           reply:=readkey;

     until reply in ['0'..'5'];

end;

procedure getData;

var

   answer :char;

   aFriend : Friend;

begin

      assign(f,'friends.dat');

      {$i-} reset(f); {$i+}

      if ioresult <>0 then

         rewrite(f);

      seek(f,filesize(f));

     {χρησιμοποιώ την εντολή seek(f, filesize(f)) για να τοποθετήσω το δείκτη εγγραφής

     μετά την τελευταία εγγραφή ώστε να προσθέσει τις εγγραφές από εκεί και πέρα,

     αλλιώς κάθε φορά που θα έτρεχα το πρόγραμμα θα έγγραφε ξανά από την

     εγγραφή 0 κτλ λόγω της reset(f)}

     clrscr;

     writeln('ΔΩΣΕ ΤΑ ΟΝΟΜΑΤΑ ΤΩΝ ΦΙΛΩΝ ΣΟΥ');

     repeat

           with aFriend do begin

                write('NAME : ');readln(name);

                write('ADDRESS : ');readln(address);

                write('PHONE : ');readln(phone);

           end;

           write(f, aFriend);

           write('an other friend ? ');readln(answer);

     until answer in ['o','O','ο','Ο'];

     close(f);

end;

procedure displayRecords;

var

   aFriend : Friend;

procedure displayline;

var

   i:byte;

begin

     for i:=1 to 50 do write('=');

     writeln;

end;

begin

     clrscr;

     assign(f,'friends.dat');

      {$i-} reset(f); {$i+}

      if ioresult =0 then begin

         writeln('ΕΜΦΑΝΙΣΗ ΟΝΟΜΑΤΩΝ ΦΙΛΩΝ');

         writeln('name':21,'address':16,'phone':7);

         displayline;

         while not eof(f) do begin

               read(f,aFriend);

               with aFriend do

                    writeln(name:21,address:16,phone:7);

               end;

         close(f);

      end

      else writeln('File Frients.dat not found');

      readln;

end;

procedure ReadAndDisplayARecord;

var

   recNo:byte;

   rec :Friend;

begin

     clrscr;

     assign(f,'friends.dat');

      {$i-} reset(f); {$i+}

      if ioresult=0 then begin

          writeln('ΑΝΑΖΗΤΗΣΗ ΕΓΓΡΑΦΗΣ ΜΕ ΒΑΣΗ ΤΟΝ ΑΡΙΘΜΟ ΕΓΓΡΑΦΗΣ');

          write('give the record number ');readln(recNo);

          seek(f, recNo);

          read(f, rec);

          with rec do

                writeln(name:21,address:16,phone:7);

          close(f);

      end

      else writeln('File Frients.dat not found');

      readln;

end;

procedure Find_changeARecordByrecNo;

var

   recNo: byte;

   rec:friend;

begin

    displayrecords;

    assign(f,'friends.dat');

    {$i-} reset(f); {$i+}

    if ioresult=0 then begin

         writeln('ΑΝΑΖΗΤΗΣΗ ΕΓΓΡΑΦΗΣ ΜΕ ΒΑΣΗ ΤΟΝ ΑΡΙΘΜΟ ΕΓΓΡΑΦΗΣ ΚΑΙ ΑΛΛΑΓΗ ΤΟΥ ΟΝΟΜΑΤΟΣ');

         write('give the record number ');readln(recNo);

         seek(f, recNo);

         read(f, rec);

         write('give an other name ?');readln(rec.name);

         seek(f,recNo);

         write(f, rec);

         close(f);

      end

      else writeln('file Frients.dat not found');

      readln;

end;

procedure Find_ChangeARecordbyaname;

var

   recNo: byte;

   rec:friend;

   onoma : string[20];

   found : boolean;

begin

     displayrecords;

     assign(f,'friends.dat');

      {$i-} reset(f); {$i+}

      if ioresult=0 then begin

         writeln('ΑΝΑΖΗΤΗΣΗ ΕΓΓΡΑΦΗΣ ΜΕ ΒΑΣΗ ΤΟ ΟΝΟΜΑ');

         writeln('ΚΑΙ ΑΛΛΑΓΗ ΠΕΡΙΕΧΟΜΕΝΟΥ ΠΕΔΙΟΥ');

         write('give the name ');readln(onoma);

         found:=false;

         while (not eof(f)) and (not found) do begin

               read(f, rec);

               if rec.name=onoma then begin

                  found:=true;

                  recno:=filepos(f)-1;

                  write('give an other name ?');readln(rec.name);

                  seek(f,recNo);writeln(recno);

                  write(f, rec);

               end

         end;

         if not found then begin

            writeln('δεν βρεθηκε εγγραφή με αυτό το όνομα και δεν έγινε καμμιά αλλαγή');

         end;

         close(f);

      end

      else writeln('File Friends.dat not found');

      readln;

end;

begin {main}

      repeat

            menu(choice);

            case choice of

                 '1' : getData;

                 '2' : displayrecords;

                 '3' : ReadAndDisplayARecord;

                 '4' :begin

                           Find_changeARecordByrecNo;

                           displayrecords;

                     end;

                 '5' : begin

                            Find_ChangeARecordbyaname;

                            displayrecords;

                       end;

            end;

      until choice='0';

end.

